DOI: 10.5281/zenodo.13371966
title: Archaic Lyric Agora (ALyrA) - Ontoterminology of Archaic Lyric Poets - version 1.0

description: This is version 1.0 of the ALyrA ontoterminology (i.e., 'a terminology whose conceptual system is a formal ontology', Roche 2007), which focuses on modelling Archaic Lyric poets (799-430 BCΕ). AlyrA stands for "Archaic Lyrical Agora". 

The ontoterminology was created by Rafail Giannadakis (Department of Philology, University of Crete) under the supervision of Dr. Maria Papadopoulou (Department of Philology, University of Crete).

We gratefully acknowledge the generous funding of the European Commission in the context of the TALOS AI for SSH project (Grant agreement ID: 101087269).

Contact: giannadakis.uni@gmail.com 
description: This version includes:

1. The .otb file. To view it, please import it in  Tedi ontoTerminology Editor

2. The .rdf file. To view it, please import it in Protégé ontology editor

3. The .html files that includes the proper names modelled

5. The .html files that includes the terms modelled

Software:

This dataset was built with Tedi ontoTerminology Editor. Tedi ontoTerminology Editor is a software platform for the creation and management of ontoterminologies. Tedi software platform is distributed by the University of Crete (free for academic use, contact: christophe.roche@uoc.gr).

The .rdf file is queryable via a SPARQL endpoint.

Downloads:

Download the RDF/OWL file; to view the ontology metrics import this file in Protégé (open access ontology editor); to visualize the ontology import this file in VOWL

To view the .html files in your browser download them and rename them using the .html extension.
description: Metrics

Tedi:



Number of concepts: 29

Number of objects: 101

Number of terms: 58

Number of proper names: 247


RDF/Protégé:



Axiom:  4,782

Logical axiom count: 2,348

Declaration axioms count: 177  

Class count: 43

Object property count: 21         

Data property count: 6 

Individual count : 610    

Annotation Property count: 36

SubClassOf: 81              

Disjoint Classes: 53

Object Property Domain: 20      

Object Property Range:  20

Data Property Domain: 6

Data Property Range: 6

Class Assertion  798      

Object Property Assertion: 1,291            

Data Property Assertion: 73

Annotation Assertion: 2,235      

Annotation Property Domain: 12            

Annotation Property Range Of: 10
description: 5 SPARQL Queries (queryable via a SPARQL endpoint):

#1 Which are the different Archaic Lyric poems and what differentiates each of them?

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX alyra: <http://www.ontologia.fr/OTB/ALyrA_v.1.0#>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>

SELECT ?poems ?difference ?definition
FROM <http://ontologia.fr/OTB/ALyrA_v.1.0.rdf>
WHERE {    
       ?x rdfs:subClassOf* alyra:Archaic_Lyric_Poem;
             rdfs:label ?poems;
             otv:ownDifference ?y;
	skos:definition ?definition.
       ?y rdfs:label ?difference
FILTER (lang(?poems) = 'en')
FILTER (lang(?definition) = 'en')
}


 
description: #2 What is an “hymn” and which Archaic Lyric poets/poetesses have composed “hymns”?

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX alyra: <http://www.ontologia.fr/OTB/ALyrA_v.1.0#>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>

SELECT ?definition ?authorName ?centuryOfLiving 
FROM <http://ontologia.fr/OTB/ALyrA_v.1.0.rdf>
WHERE {    
    ?x rdfs:label "hymn"@en.
    ?x skos:definition ?definition.
          FILTER(langMatches(lang(?definition), "en"))
    ?y otv:instanceOf ?x;
          alyra:isWrittenBy ?author.
    ?author rdfs:label ?authorName;
            alyra:centuryOfLiving ?c.
    ?c rdfs:label ?centuryOfLiving.
}
ORDER BY ?centuryOfLiving


 
description: #3 Who was Pindar?

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX alyra: <http://www.ontologia.fr/OTB/ALyrA_v.1.0#>

SELECT ?propertyName ?value
FROM <http://ontologia.fr/OTB/ALyrA_v.1.0.rdf>
WHERE {
  	alyra:pindarus_boeotus ?x ?y .
  	?x rdfs:label ?propertyName .
  	?y rdfs:label ?value .
  	FILTER(LANG(?value) = 'en')
}
ORDER BY ?value


 

 
description: #4 Which is the four-digit "Thesaurus Linguae Graecae" identifying number of each Archaic Lyric composer?

 

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX alyra: <http://www.ontologia.fr/OTB/ALyrA_v.1.0#>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>

SELECT ?nameOfComposer ?tljNumber
FROM <http://ontologia.fr/OTB/ALyrA_v.1.0.rdf>

WHERE {
?x rdfs:subClassOf* alyra:Archaic_Lyric_Composer.
?y otv:instanceOf* ?x.
?y rdfs:label ?nameOfComposer;
   	alyra:tlgIdentifyingNumber ?tljNumber}

 
description: #5 What is the place where most Ancient Greek lyric composers were born and who were they?

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX otv: <http://www.ontologia.fr/OTB/otv#>
PREFIX alyra: <http://www.ontologia.fr/OTB/ALyrA_v.1.0#>

SELECT ?place ?coordinates ?poetLabel ?resourcePoet
FROM <http://ontologia.fr/OTB/ALyrA_v.1.0.rdf>

WHERE {
  {
    SELECT ?place (COUNT(?y) AS ?count)
    WHERE {
      ?x rdfs:subClassOf* alyra:Archaic_Lyric_Composer.
      ?y otv:instanceOf ?x.
      ?y alyra:bornIn ?p.
      ?p rdfs:label ?place.
    }
    GROUP BY ?place
    ORDER BY DESC(?count)
    LIMIT 1
  }
  ?x rdfs:subClassOf* alyra:Archaic_Lyric_Composer.
  ?y otv:instanceOf ?x.
  ?y alyra:bornIn ?p.
  ?p rdfs:label ?place.
  OPTIONAL { ?p alyra:pleiadesCoordinates ?coordinates. }
  ?y rdfs:label ?poetLabel.
  OPTIONAL {
    SELECT ?y (SAMPLE(?resource) AS ?resourcePoet)
    WHERE {
      ?y rdfs:seeAlso ?resource.
      FILTER(CONTAINS(LCASE(STR(?resource)), "perseus:author"))
    }
    GROUP BY ?y
  }
}
ORDER BY ?poetLabel